
Perhaps the best way to learn how to use NCL is to write a couple of NCL procedures
and execute them.  This section guides you through the steps required to do so.  This
section covers the following topics:

What you need to do before you start using NCL

Getting started with NCL procedures

Getting started with functions

Writing a MSGPROC NCL procedure

What to do after completing this tutorial

This section has two purposes:  to lead you through the mechanical tasks of creating,
checking, and executing NCL procedures;  and to give you a feeling for the language.

Explaining the mechanics means explaining how to use Edit Services to create and
check NCL procedures.  It also means explaining how to use Operator Control
Services (OCS) to execute NCL procedures.

The basic building blocks of the language are procedures, functions, core statements,
verbs, variables, expressions, labels, and comments.  This section does not discuss the
details of the language and it certainly does not cover the whole language, but it does
look at what is important to get you started using NCL.

Before You Start Let us begin at the beginning.  Before you can write anything in NCL, there are a
couple of important things you should check:

1. You must be able to log on to your NonStop NET/MASTER Management Services
(MS) system.

You must have a user ID definition record in User ID Management Services
(UMS).  It must have a valid user ID and password.  You must know how to log
on to your NonStop NET/MASTER MS system.

If you do not have a user ID definition record, refer to the NonStop NET/MASTER
MS System Management Guide or contact your system manager for assistance.  If
you do not know how to log on to your NonStop NET/MASTER MS system, refer
to the NonStop NET/MASTER MS Operator’s Guide for step-by-step instructions.

2. You must be able to use Edit Services.

Your user ID definition record in UMS must give you the authority to select Edit
Services from the NonStop NET/MASTER MS primary menu.  You must know
how to choose options from a menu.  The editor used by Edit Services is PS Text
Edit (TEDIT);  you must know how to use TEDIT.

If your user ID definition record does not give you the authority to select Edit
Services, refer to the NonStop NET/MASTER MS System Management Guide or
contact your system manager for assistance.  Refer to the NonStop NET/MASTER
MS Operator’s Guide for information on using Edit Services.  If you do not know
how to use TEDIT, refer to the PS Text Edit and PS Text Format User’s Guide and the
PS Text Edit Reference Manual.

Note If you do not have the authority to select Edit Services from the NonStop NET/MASTER MS primary
menu, you cannot use the EDIT command to invoke Edit Services.  Refer to the NonStop NET/MASTER
MS Operator’s Guide for information on using the EDIT command.  The complete syntax of the EDIT
command is described in the NonStop NET/MASTER MS Command Reference Manual.

3. You must be able to use Operator Control Services (OCS).

Your user ID definition record in UMS must give you the authority to select OCS
from the NonStop NET/MASTER MS primary menu.  You must know how to
enter NonStop NET/MASTER MS commands from the OCS command input line.

If your user ID definition record does not give you the authority to select OCS,
refer to the NonStop NET/MASTER MS System Management Guide or contact your
system manager for assistance.  If you do not know how to enter NonStop
NET/MASTER MS commands from the OCS command input line, refer to the
NonStop NET/MASTER MS Operator’s Guide for details.



Before You Start

An NCL Tutorial

106160 Tandem Computers Incorporated 3–3

4. You must have a library, a user procedure library, in which to store your NCL
source files when you create them.

More than one user can use the same user procedure library.  The name of your
library is defined either by the PARAM command or in your user ID definition
record.

Refer to the NonStop NET/MASTER MS System Management Guide for information
on PARAM commands and defining a user procedure library using UMS.  The
syntax of the PARAM command is described in the NonStop NET/MASTER MS
Command Reference Manual.

5. You must have an authority level of at least 2 (to execute the PROFILE MSGPROC
command described later in this section).

If your user ID definition record does not give you the authority to execute the
PROFILE MSGPROC command, refer to the NonStop NET/MASTER MS System
Management Guide or contact your system manager for assistance.

Contact your system manager for assistance if you need additional help verifying
these checks.

You may find it convenient, whenever you enter OCS during this tutorial, to set test
mode on by setting PROFILE NCLTEST equal to YES.  Test mode is usually set on
when you are developing NCL procedures.

Note After you have verified these checks, the tutorial takes about three hours to complete.

Getting Started With
NCL Procedures

Here is a step-by-step example that shows you how to create, test compile, correct, and
execute your first NCL procedure.  The name of the NCL procedure is HELLO1.

When executed, this NCL procedure displays the following message in the OCS
message display area:

 hello, world

The HELLO1 NCL procedure introduces the language constructs of core statements,
labels, and expressions.

Note In this step-by-step example, each screen shows what you type and the results in bold type.

Your First NCL
Procedure—HELLO1

When using your NonStop NET/MASTER MS system, your starting point is the
NonStop NET/MASTER MS primary menu.  It is assumed that you are logged on to
your NonStop NET/MASTER MS system and have the primary menu in front of you,
as shown in the following screen:

 SYS1------- NonStop NET/MASTER MS T6581D30 - 31OCT94 : Primary Menu -------SYS1
 Select Option ===>

    2   - Operator Control Services                              Userid NMTMG
    8   - Edit Services                                          Term   #7493572
    E   - Console Extras                                         Time   10:59:00
    M   - User ID Management Services                            FRI 11-NOV-1994
    S   - System Support Services                                OPSYS  GUARDIAN
    U   - User Services                                          Window 1
    X   - Terminate Window.Exit

                  Last logged on  FRI 11-NOV-1994    07:39
                  at terminal     $ATPG.#TERM4

The NonStop NET/MASTER MS Operator’s Guide describes the components of the
NonStop NET/MASTER MS primary menu.

Step 1

Select Edit Services from the NonStop NET/MASTER MS primary menu.

Type 8 after the Select Option ===> prompt, as shown in the following screen, and
then press the ENTER key:

 SYS1------- NonStop NET/MASTER MS T6581D30 - 31OCT94 : Primary Menu -------SYS1
 Select Option ===> 8

    2   - Operator Control Services                              Userid NMTMG
    8   - Edit Services                                          Term   #7493572
    E   - Console Extras                                         Time   10:59:36
    M   - User ID Management Services                            FRI 11-NOV-1994
    S   - System Support Services                                OPSYS  GUARDIAN
    U   - User Services                                          Window 1
    X   - Terminate Window.Exit

                  Last logged on  FRI 11-NOV-1994    07:39
                  at terminal     $ATPG.#TERM4

You use Edit Services in NonStop NET/MASTER MS to create NCL procedures and
panels.  Edit Services uses TEDIT.

Step 2

Type the file name of the new NCL procedure.  This is also the name of a new TEDIT
file.

When you enter Edit Services, your cursor is positioned after the File Spec: prompt.
Type the file name, HELLO1, after the File Spec: prompt, as shown in the following
screen, and then press the ENTER key:

 ------------------------- Edit Services : Selection List ----------------------
  File Spec: HELLO1
  Subvolume: $DATA9.MYNCLLIB
                      T=Test a panel  S/B=Browse  C=Compile  D=Delete  E=Edit

         Filename Code        EOF Last Modification    Owner   RWEP

   F3=Exit    F7=Prev    F8=Next    F11=Error List

You can type HELLO1 in either uppercase or lowercase.  The case is not significant in
a file name.  When you press the ENTER key after typing the name of the new NCL
procedure, you invoke TEDIT.

Since this is your first NCL procedure, the Edit Services selection list screen does not
display any NCL procedure names when you enter Edit Services.  This section does
not describe in detail the components of the Edit Services selection list screen;  the
components are described in the NonStop NET/MASTER MS Operator’s Guide.

Step 3

Type the new NCL procedure into a TEDIT file, in uppercase and lowercase letters, as
shown in the following screen:

 hello_world: PROCEDURE
    SAY "hello, world"
 END hello_world

   1) $DATA9.MYNCLLIB.HELLO1 1/24 (BOF) (EOF) 1:79 Insert    D30.  BLOCK

The screen is blank when you invoke TEDIT because the NCL procedure is initially an
empty TEDIT file called HELLO1.  As you type, your input is entered in the TEDIT
file.  The usual TEDIT indicators appear at the bottom of the screen while you are
using TEDIT.  If you do not know how to use TEDIT, refer to the PS Text Edit and PS
Text Format User’s Guide and the PS Text Edit Reference Manual.

Step 4

Save the new NCL procedure and exit TEDIT.

Press the F1 function key and then type EXIT at the Command: prompt, as shown in
the following screen (you can also press the SHIFT/F16 function key combination to
exit TEDIT):

 Command: EXIT
    SAY "hello, world"
 END hello_world

   1) $DATA9.MYNCLLIB.HELLO1 1/24 (BOF) (EOF) 1:79 Insert    D30.  BLOCK

When you press the F1 function key, the Command: prompt hides the top line of your
new NCL procedure.  After typing EXIT, you may be asked the following question:

Do you really want to exit? Respond Y or N:

Whether this question appears depends on how your TEDIT profile is configured.  If
you see this question, answer Y.

When you save the new NCL procedure and exit TEDIT, you return to the Edit
Services selection list screen.  The name of the new NCL procedure, and other
information, is now automatically displayed on the screen.  (When the number of NCL
procedures exceeds the number that can be displayed on a single screen, you can use
the F7 and F8 function keys to page backward and forward between multiple screens.)

Step 5

Perform a test compilation of the new NCL procedure.

Move the cursor before HELLO1 using the arrow keys or the TAB key, type the letter
C, as shown in the following screen, and then press the ENTER key:

 ------------------------- Edit Services : Selection List ----------------------
  File Spec:
  Subvolume: $DATA9.MYNCLLIB
                      T=Test a panel  S/B=Browse  C=Compile  D=Delete  E=Edit

         Filename Code        EOF Last Modification    Owner   RWEP
    C    HELLO1    101       2132 11-NOV-1994 11:03:11 100,21  CCCC

   F3=Exit    F7=Prev    F8=Next    F11=Error List

Performing a test compilation on an NCL procedure checks the syntax of the NCL
procedure without executing it.  It is good practice to perform a test compilation of an
NCL procedure before you attempt to execute it.  Choosing C in Edit Services to check
an NCL procedure has the same result as using the NCLCHECK command in OCS.

The screen also shows the meaning of the other letters—T, S, B, D, and E—that you can
place before a file name.

Step 6

Check for mistakes after the test compilation by looking at the message line that
appears at the top of the Edit Services selection list screen, as shown in the following
screen:

 ------------------------- Edit Services : Selection List ----------------------
  File Spec:
  Subvolume: $DATA9.MYNCLLIB
                      T=Test a panel  S/B=Browse  C=Compile  D=Delete  E=Edit
 *END* 0 ERRORS 0 WARNINGS
         Filename Code        EOF Last Modification    Owner   RWEP
         HELLO1    101       2132 11-NOV-1994 11:03:11 100,21  CCCC

   F3=Exit    F7=Prev    F8=Next    F11=Error List

There are three types of mistakes that can result from a test compilation.  From least to
most severe these are warnings, errors, and fatal errors.

Note A test compilation does not check the logic of an NCL procedure.  Accordingly, if a test compilation is
successful, it does not guarantee that the NCL procedure will execute as expected.  You can use the
DEBUG commands, discussed in Section 9, “Debugging an NCL Process,” to detect errors in logic.  The
complete syntax of the DEBUG commands is described in the NonStop NET/MASTER MS Command
Reference Manual.

Successful Test Compilation.  If you typed the HELLO1 NCL procedure exactly as shown
earlier in this section, then you will have 0 errors and 0 warnings.  This means that you
have made no syntax errors; the test compilation was successful.  If the test
compilation was successful, go to Step 7.

Unsuccessful Test Compilation.  If you made a mistake typing the HELLO1 NCL
procedure, then you will have some errors or some warnings.  This means that you
have made some syntax errors; the test compilation was unsuccessful.  If the test
compilation was unsuccessful, go to Step 11.

Step 7

Note If your test compilation in Step 6 was unsuccessful, you should proceed directly to Step 11.

Exit Edit Services and return to the NonStop NET/MASTER MS primary menu by
pressing the F3 function key.

Step 8

Select Operator Control Services (OCS) from the NonStop NET/MASTER MS primary
menu.

Type 2 after the Select Option ===> prompt, as shown in the following screen, and
then press the ENTER key:

 SYS1------- NonStop NET/MASTER MS T6581D30 - 31OCT94 : Primary Menu -------SYS1
 Select Option ===> 2

    2   - Operator Control Services                              Userid NMTMG
    8   - Edit Services                                          Term   #7493572
    E   - Console Extras                                         Time   11:10:18
    M   - User ID Management Services                            FRI 11-NOV-1994
    S   - System Support Services                                OPSYS  GUARDIAN
    U   - User Services                                          Window 1
    X   - Terminate Window.Exit

                  Last logged on  FRI 11-NOV-1994    07:39
                  at terminal     $ATPG.#TERM4

After creating your new NCL procedure and performing a successful test compilation
in Edit Services, you must select OCS to execute it and to see the results.

Step 9

Execute the HELLO1 NCL procedure.

When you enter OCS, your cursor is positioned after the OCS command input line
prompt (=>).  Type the NonStop NET/MASTER MS command START HELLO1 after
the prompt, as shown in the following screen, and then press the ENTER key:

  (11:15) --------------------- OPERATOR CONTROL SERVICES ----------------------

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=> START HELLO1

Note The letter M before the OCS command input line prompt is called a mode indicator and shows that you
are a monitor-class user (allowed to receive monitor-class messages).

You can also execute the NCL procedure by typing EXEC HELLO1 or just HELLO1.  If
you just type HELLO1 and you receive the message NNM0201 INVALID COMMAND
'HELLO1', the setting of the SYSPARMS AUTOEXEC command is equal to NO or
OPSYS.  If you receive this message, type the command START HELLO1 instead.
Refer to the NonStop NET/MASTER MS Command Reference Manual for a complete
description of the SYSPARMS AUTOEXEC command.

Step 10

Check the results from executing the HELLO1 NCL procedure.

The results are displayed in the roll-delete message area of the OCS window, as shown
in the following screen:

  (11:16) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START HELLO1
   hello, world
   NNM1005 START HELLO1 PROCESSING COMPLETE. NCLID 001034
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

This screen displays the following information:

The first line of the results—START HELLO1—shows the command you used to
execute the NCL procedure.

The second line—hello, world—shows the output from the NCL procedure.
This means you have successfully executed your first NCL procedure.

The third line—NNM1005 …— indicates that NonStop NET/MASTER MS has
finished processing the START command.  The number 001034 in this line is called
an NCL process identifier (NCL ID).  Every NCL procedure is automatically
assigned a unique NCL ID by NonStop NET/MASTER MS when the NCL
procedure begins execution.

Note After this step, you should skip Steps 11 through 16 and go to the subsection “Features of HELLO1” later
in this section.

Step 11

Note This step follows Step 6 if your test compilation was unsuccessful.  If you have just completed Step 10,
you should go to the subsection “Features of HELLO1” later in this section.

Note the number of mistakes by looking at the message line that appears at the top of
the Edit Services selection list screen, as shown in the following screen:

 ------------------------- Edit Services : Selection List ----------------------
  File Spec:
  Subvolume: $DATA9.MYNCLLIB
                      T=Test a panel  S/B=Browse  C=Compile  D=Delete  E=Edit
 *END* 1 ERRORS 0 WARNINGS
         Filename Code        EOF Last Modification    Owner   RWEP
         HELLO1    101       2132 11-NOV-1994 11:03:11 100,21  CCCC

   F3=Exit    F7=Prev    F8=Next    F11=Error List

Assume you left out the closing quote (”) after the word “world.”  The message line
shows that you have made 1 error and have no warnings.  Other common errors are
misspelling and adding or omitting nonalphanumeric characters, such as colons (:) or
semicolons (;).



Getting Started With NCL Procedures

An NCL Tutorial

106160 Tandem Computers Incorporated 3–15

Step 12

Examine the error list created by the test compilation of the NCL procedure to
determine the cause of the error.

Press the F11 function key to display the error list.  The error list is shown in the Edit
Services NCL compile report screen, as shown in the following screen:

 ------------------------- Edit Services : NCL Compile Report ------------------
 Page Down for more lines
 HELLO1 3 END hello_world
 HELLO1 3             !
 FATAL ERROR: OPEN STRING AT EOF
 *END* 1 ERRORS 0 WARNINGS

      F3=Exit    F7=Prev    F8=Next

Notice the error message OPEN STRING AT EOF.  This means you used an opening
quote but omitted its corresponding closing quote.  As the message indicates, this is a
fatal compilation error.

The error list can display up to four screens of errors.

Step 13

Exit the Edit Services NCL compile report screen and return to the Edit Services
selection list screen by pressing the F3 function key.

Step 14

Select the HELLO1 NCL procedure for editing.

Move the cursor before HELLO1 using the arrow keys or the TAB key, type the letter
E, as shown in the following screen, and then press the ENTER key:

 ------------------------- Edit Services : Selection List ----------------------
  File Spec:
  Subvolume: $DATA9.MYNCLLIB
                      T=Test a panel  S/B=Browse  C=Compile  D=Delete  E=Edit

         Filename Code        EOF Last Modification    Owner   RWEP
    E    HELLO1    101         84 11-NOV-1994 11:03:11 100,21  CCCC

   F3=Exit    F7=Prev    F8=Next    F11=Error List

When you press the ENTER key after typing the letter E next to a file name, you
invoke TEDIT and open the file you select for editing.

Step 15

Examine the NCL procedure and insert the missing closing quote (”) after the letter
“d” in the word “world,” as shown in the following screen:

 hello_world: PROCEDURE
    SAY "hello, world" <-- Insert closing quote here
 END hello_world

   1) $DATA9.MYNCLLIB.HELLO1 1/24 (BOF) (EOF) 1:79 Insert    D30.  BLOCK

Step 16

Follow Steps 4 through 10 described earlier in this section:

Step 4—save the NCL procedure and exit TEDIT.

Step 5—perform a test compilation of the NCL procedure.

Step 6—check for mistakes after the test compilation.

Step 7—exit Edit Services and return to the NonStop NET/MASTER MS primary
menu.

Step 8—select Operator Control Services from the NonStop NET/MASTER MS
primary menu.

Step 9—execute the HELLO1 NCL procedure.

Step 10—check the results from executing the HELLO1 NCL procedure.

Features of HELLO1 Now that you have successfully executed your first NCL procedure, it is instructive to
look at some of its features.

The File Name—HELLO1

HELLO1 is the name of a TEDIT file containing NCL statements.  It is called the NCL
procedure source file name.

It is also the name of the external procedure you have just created.  An external name
is the name used by other source files to refer to an NCL procedure.  The Edit Services
selection list screen shows the name of the NCL procedure you have just created under
the heading Filename.

The HELLO_WORLD: Label

HELLO_WORLD: at the beginning of the NCL procedure is called a label.  It labels an
internal procedure.  An internal name is the name used from within a source file to
refer to an NCL procedure.  Labels are normally used as the target of other statements
such as CALL, GOSUB, and SIGNAL LABEL.  The last character of a label in this
context must be a colon (:).

The PROCEDURE Core Statement

PROCEDURE is a core statement.  It marks the beginning of the NCL procedure.  It is
used to explicitly declare an NCL procedure.  Every PROCEDURE statement must
have a corresponding END statement that terminates the NCL procedure declaration.

The SAY Core Statement

SAY is a core statement.  It must be followed by an expression.  In this example, the
expression is “hello, world.”  SAY displays the expression on an OCS screen.

The Quoted String—“hello, world”

“hello, world” is a lexical element called a quoted string.  The characters between the
opening and closing quotes of a quoted string are constant.  They do not change
during the execution of an NCL process.

The END Core Statement

END is a core statement.  It marks the end of the NCL procedure.  Every PROCEDURE
statement must have a matching END statement.

The HELLO_WORLD Label

HELLO_WORLD at the end of the NCL procedure is a label.  Including it after the
END core statement is not required but is good practice.  It lets the compiler check that
the label at the end of the NCL procedure matches the label at the beginning of the
NCL procedure.  When a label follows END, it is not followed by a colon (:).

The HELLO2 NCL
Procedure

Now let us create another NCL procedure called HELLO2, as shown in the following
listing:

hello_world: PROCEDURE
   /* Displays "hello, world" on an OCS screen */
   /* using the WRITE verb                     */
   &1 = "hello,"
   &2 = world
   WRITE DATA=&1 &2
END hello_world

Follow the steps described earlier in this section to create, test compile, correct, and
execute HELLO2.  When executed, the NCL procedure displays the following message
in the OCS message display area:

 hello, world

HELLO2 illustrates the point that there are often many ways to produce the same
result using NCL.  HELLO2 introduces the language constructs of verbs, variables,
and comments.

Features of HELLO2 Here are some of the features of the NCL procedure HELLO2.

The Comment Statement—/* … */

HELLO2 has a statement called a comment statement (usually called a comment) after
the label HELLO_WORLD: and the PROCEDURE core statement.

Comments are used to describe what an NCL procedure does at a certain point.  Any
characters between the comment delimiters (/* and */) are ignored by the compiler.
Comments can be used anywhere that a blank can be used in an NCL procedure.  (The
compiler replaces a comment by a blank during compilation.)

The Variables—&1 and &2

HELLO2 contains variables.  A variable is a lexical element containing data that can
change during the execution of an NCL process.  All variables in NCL (except input
field variables in panel description files) begin with an ampersand (&).

NCL has five types of variables:  simple, complex, stem, compound, and nested.  The
variables &1 and &2 are called simple variables;  simple variables provide direct access
to data and have only one ampersand.

The Quoted String—“hello,”

“hello,” is a quoted string.  It is surrounded by quotes because the last character is a
comma (,).  Without quotes, NCL assumes that a comma at the end of a line is a line
continuation character.  This is the character you use at the end of a line in a TEDIT
file when you want to continue a single NCL statement to the next line.

The Unquoted String—world

Quotes are not needed for “world.”  It has only alphabetic characters.  It is another
type of lexical element called an unquoted string.

The Assignment Statements

HELLO2 has two assignment statements.  An assignment statement assigns the value
on the right side of an equal sign (=) to the variable on the left side of the equal sign.

In HELLO2, the first assignment statement assigns the value of “hello,” to &1 and the
second assignment statement assigns the value of “world” to &2.

The WRITE Verb

WRITE is a verb.  DATA is a verb keyword.  The DATA keyword must be followed by
an expression containing the text of a message.  The expression can be constant or
variable;  in HELLO2, the DATA keyword is followed by a variable expression.

WRITE can do many things with the expression following DATA.  One thing WRITE
can do is send the expression following DATA to the message display area of an OCS
window.  This function of WRITE is identical to the function of SAY.  The HELLO3
NCL procedure later in this section explores some more functions of WRITE.

The HELLO3 NCL
Procedure

The final HELLO NCL procedure we will create is called HELLO3, as shown in the
following listing:

hello_world: PROCEDURE
   &1 = "hello,"
   &2 = world
   WRITE HLITE=BLINK,
         COLOR=RED,
         INTENS=HIGH,
         ALARM=YES,
         NRD=OPER,
         DATA=&1 &2
END hello_world

Follow the steps described earlier in this section to create, test compile, correct, and
execute HELLO3.  When executed, this NCL procedure displays the following
message in the OCS message display area:

 hello, world

The WRITE verb changes the attributes of the message to enhance its appearance.
HELLO3 illustrates the power of the WRITE verb.

Features of HELLO3 Here are some of the features of the NCL procedure HELLO3.

Changing Message Appearance Using the WRITE Verb

The WRITE verb can change the appearance of a message with keywords such as
HLITE, COLOR, and INTENS and qualifiers such as BLINK, RED, and HIGH.  By
using these keywords and qualifiers with the WRITE verb, HELLO3 displays a red,
blinking message in high intensity.

Note The exact appearance of a message when it is displayed depends on your terminal type.  For example,
Tandem 6526 and 6530 terminals restrict the range of displayable colors and 6526 terminals display a
blinking message in italics.

Changing Internal Message Attributes Using the WRITE Verb

The WRITE verb can change some of the internal characteristics of a message.  Using
the ALARM keyword with the YES qualifier causes the message to beep when it is
displayed.  Using the NRD keyword with the OPER qualifier turns the message into a
non-roll-delete (NRD) message with the OPER attribute.

For the complete syntax of the WRITE verb, refer to the NonStop NET/MASTER NCL
Reference Manual.

Note To delete the message “hello, world,” (since it is a NRD message with the OPER attribute) you must use
NRD cursor deletion.  Move the cursor to the line on which the message is displayed and press the
ENTER key.

Getting Started With
Functions

The examples in this section have focused on NCL procedures.  This subsection
discusses functions.

An NCL procedure is a sequence of NCL statements that performs a task and,
optionally, returns a result.  For example, an NCL procedure can display a message
such as “hello, world” in various ways.

Functions also perform tasks, but do so in a much more limited way than NCL
procedures.  Functions are designed for one specific kind of task:  taking a list of
values (called arguments or parameters) and returning a single value.  The value is
used to evaluate the expression from which the function is called.

When you call a function, you use its name, optionally followed by a list of parameters
that must be enclosed in parentheses (( )).  You must be careful not to include a space
between the name of the function and the opening parenthesis (();  if you do, NCL
does not recognize the function.  Parameters in the list are separated by commas (,).
Spacing between parameters in the parameter list is not important.

NCL is distributed with a range of functions, called built-in functions.  You can also
write your own functions using NCL.  These are called user-written functions.  This
subsection introduces built-in functions and then looks at user-written functions.

Refer to the NonStop NET/MASTER NCL Reference Manual for a complete description of
all the built-in functions distributed with NonStop NET/MASTER MS.

The BUILTIN1 NCL
Procedure

The first NCL procedure we will create uses a single built-in function.  The name of the
NCL procedure is BUILTIN1;  the name of the built-in function is SUBSTR.  BUILTIN1
is shown in the following listing:

builtin1: PROCEDURE
   WRITE DATA=SUBSTR( "hello, world", 1, 5 )
END builtin1

Follow the steps described earlier in this section to create, test compile, correct, and
execute BUILTIN1.  When executed, this NCL procedure displays the following
message in the OCS message display area:

  hello

The BUILTIN1 NCL procedure uses the WRITE verb to illustrate the use of the
SUBSTR built-in function.

Features of BUILTIN1 Here are some of the features of the BUILTIN1 NCL procedure.

The SUBSTR Built-in Function

SUBSTR is the name of a built-in function used to extract a substring from a string of
characters.

In BUILTIN1, three parameters are passed to the SUBSTR built-in function.  The first
parameter is the string from which to extract the substring:  in this case,

“hello, world.”  The second parameter is a number, indicating the character at which
to begin the extraction—1, in this case, means the first character of the string, which is
the letter “h” in this example.  The third parameter is also a number, indicating the
number of characters to extract from the string—5, in this case, means five characters,
which are the letters “h,” “e,” “l,” “l,” and “o” in this example.

Displaying the Results of a Function Using the WRITE Verb

Recall that the DATA keyword of the WRITE verb must be followed by an expression
containing the text of a message.  In addition, recall that functions are called from
expressions and that the result of a function is used as part of expression evaluation.

When NCL encounters an expression in a statement, it evaluates the expression and
then uses the result.  In BUILTIN1, the result of the built-in function is the substring
“hello.”  Accordingly, the WRITE verb displays “hello” in the OCS message display
area.

The BUILTIN2 NCL
Procedure

NCL has many built-in functions that you can use to perform a variety of operations.
To illustrate the use of some more built-in functions, let us a create an NCL procedure
called BUILTIN2, as shown in the following listing:

builtin2: PROCEDURE
   WRITE DATA="Builtin ABBREV: "ABBREV( "hello, world", he, 2 )
   WRITE DATA="Builtin C2X:    "C2X( "hello, world" )
   WRITE DATA="Builtin CENTER: "CENTER( "hello, world", 24, "*" )
   WRITE DATA="Builtin LENGTH: "LENGTH( "hello, world" )
   WRITE DATA="Builtin UPPER:  "UPPER( "hello, world" )
END builtin2

Follow the steps described earlier in this section to create, test compile, correct, and
execute HELLO2.  When executed, the NCL procedure displays the following
messages in the OCS message display area:

   Builtin ABBREV: 1
   Builtin C2X:    68656C6C6F2C20776F726C64
   Builtin CENTER: ******hello, world******
   Builtin LENGTH: 12
   Builtin UPPER:  HELLO, WORLD

BUILTIN2 performs various operations on “hello, world.”

Features of BUILTIN2 Here are some of the features of the BUILTIN2 NCL procedure.

The ABBREV Built-in Function

The ABBREV built-in function performs a test of whether a string of characters is a
valid abbreviation of another string of characters.  The result is a Boolean value:  either
1 if the test is TRUE or 0 if the test is FALSE.  In BUILTIN2, the result of the test is 1
(TRUE) because the string “he” matches the first two characters of the string “hello,
world.”

The C2X Built-in Function

The C2X built-in function converts any string to its hexadecimal equivalent.  In
BUILTIN2, the string “hello, world” is converted to its hexadecimal equivalent.

The CENTER Built-in Function

The CENTER built-in function centers a string of characters inside another string of
characters.  In BUILTIN2, the string of characters to center is “hello, world.”  The next
parameter specifies the length of the final string in which “hello, world” is centered.
The final parameter in this example specifies a pad character;  this character pads the
left side and the right side of “hello, world” with asterisks (*).

The LENGTH Built-in Function

The LENGTH built-in function returns the length of a specified string of characters.  In
BUILTIN2, the length of the string “hello, world” is 12.  The built-in function does not
include the opening and closing quotes as part of the string.

The UPPER Built-in Function

The UPPER built-in function converts lowercase letters to uppercase letters.  In
BUILTIN2, the result of the task is “HELLO, WORLD.”

The USERW1 NCL
Procedure

The appearance and structure of a user-written function is much the same as an NCL
procedure:  the method of declaration is very similar and it is called in much the same
way.  The following listing shows a user-written function called CUBE;  it is part of an
NCL procedure called USERW1:

userw1: PROCEDURE
   WRITE DATA="Cube of 2 is: "CUBE()
   cube: FUNCTION
      RETURN(2**3)
   END cube
END userw1

Follow the steps described earlier in this section to create, test compile, correct, and
execute USERW1.  When executed, this NCL procedure displays the following
message in the OCS message display area:

   Cube of 2 is: 8

Features of USERW1 Here are some of the features of the USERW1 NCL procedure.

The Function Call—CUBE()

CUBE() calls the user-written function CUBE.  It does not pass any parameters to
CUBE because the parameter list is empty.  Even though the parameter list is empty,
you must follow the name of the function by parentheses (( )).

CUBE executes the NCL statements in the function declaration.  The purpose of CUBE
is to produce the cube of a number.  There is no built-in function that performs this
task.

The CUBE: Label

You must name a user-written function with a label.  “CUBE:” at the beginning of the
function is a label.  The label is used to call the function.

The FUNCTION Core Statement

FUNCTION marks the beginning of the user-written function.  It is used to explicitly
declare a function.  Every FUNCTION statement must have a corresponding END
statement that terminates the function declaration.

The CUBE user-written function is called a nested function because its declaration is
inside the declaration of an NCL procedure.  You can also nest NCL procedures inside
functions, NCL procedures inside other NCL procedures, and functions inside other
functions.  You can nest NCL procedures and functions to a maximum of 256 levels.

The RETURN Core Statement

A function must have a RETURN statement, returning the result of the function to the
expression from which the function was called.  The RETURN statement in a function
must enclose the result in parentheses (( )).

In USERW1, the result of the function is the value of the expression (2**3).  2 is a
number, the double asterisks (**) represent the exponentiation operator, and the
number 3 is the power to which to raise the number 2.  The result is 8.

In USERW1, the result of the function is returned to the expression following the
DATA keyword of the WRITE verb.

The END Core Statement

END marks the end of the user-written function.  Every FUNCTION statement must
have a matching END statement.

The CUBE Label

CUBE at the end of the function is a label.  Including it after the END core statement is
not required but is good practice.  It lets the compiler check that the label at the end of
the function matches the label at the beginning of the function.  When a label follows
END, it does not require the colon (:).

The USERW2 NCL
Procedure

The CUBE user-written function produces the cube of the number 2.  With some minor
changes, however, we can create a much more useful function.  Let us create a new
NCL procedure, USERW2, with a new user-written function, called CUBE, to produce
the cube of any number, as shown in the following listing:

userw2: PROCEDURE
   WRITE DATA="Cube of "&1" is: "CUBE(&1)
   cube: FUNCTION
      RETURN(&1**3)
   END cube
END userw2

Follow the steps described earlier in this section to create, test compile, correct, and
execute USERW1 with one exception:  execute USERW2 three times using the
following three commands:

START USERW2 5
START USERW2 11
START USERW2 17

To do so, follow these steps:

1. Type START USERW2 5 at the OCS command input line and press the ENTER
key.

2. Type START USERW2 11 at the OCS command input line and press the ENTER
key.

3. Type START USERW2 17 at the OCS command input line and press the ENTER
key.

The following screen shows the result of executing the USERW2 NCL procedure three
times in the OCS message display area:

  (14:45) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START USERW2 5
   Cube of 5 is: 125
   NNM1005 START USERW2 PROCESSING COMPLETE. NCLID 008503
   START USERW2 11
   Cube of 11 is: 1331
   NNM1005 START USERW2 PROCESSING COMPLETE. NCLID 008504
   START USERW2 17
   Cube of 17 is: 4913
   NNM1005 START USERW2 PROCESSING COMPLETE. NCLID 008505
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Features of USERW2 Here are some of the features of the USERW2 NCL procedure.

The START USERW2 Command

When you execute an NCL procedure by using the START command, you can pass
parameters to it.  The START commands used in this example pass the numbers 5, 11,
and 17, respectively, to the USERW2 NCL procedure.  In each case the number is the
first (and only) parameter passed and so it is placed in the parameter variable &1.

The Parameter Variable &1

The most instructive thing to notice about USERW2 and CUBE is where each uses the
parameter variable &1.  The parameter is first used as a simple variable after the
DATA keyword in the WRITE verb between two quoted strings.  It is then used in a
parameter list when the CUBE user-written function is called.  Finally, it is used as a
term in an expression in the RETURN statement.

The final result of the CUBE user-written function in this example depends on the
value of the parameter passed to it.  If 5 is passed, the result is 125; if 11, 1331;  and if
17, 4913.  The result in each case is passed back to the WRITE verb, which displays the
value of the parameter and the result of cubing the number in the OCS message
display area.

Writing a MSGPROC
NCL Procedure

One of the most important features of NCL is its ability to intercept and process
message flows that travel to, from, and throughout NonStop NET/MASTER MS.
NonStop NET/MASTER MS is distributed with examples of the three system-level
NCL procedures that handle event messages:

EMSPROC, which is an NCL procedure that intercepts and processes Event
Management Services (EMS) messages.

LOGPROC, which is an NCL procedure that intercepts and processes messages
destined for the NonStop NET/MASTER MS activity log.

MSGPROC, which is an NCL procedure that intercepts and processes messages
destined for an OCS window.

Based on the criteria you specify in the procedure, these NCL procedures can delete a
message, modify one or more of the attributes of a message, or pass a message on
unchanged to its destination.  Each procedure uses verbs that are specific to the
function of the procedure to process a specific message type:

EMSPROC uses verbs starting with EMS to process messages.
LOGPROC uses verbs starting with LOG to process messages.
MSGPROC uses verbs starting with MSG to process messages.

While you are working in OCS, you have an OCS profile associated with your OCS
window.  Your OCS profile determines whether you have a MSGPROC procedure that
intercepts and processes messages destined for your OCS window.  (It also specifies
the name of the MSGPROC procedure if its name is different from “MSGPROC.”)
Because you can tailor aspects of your OCS profile to suit your own requirements, this
subsection briefly discusses how to write a MSGPROC NCL procedure.

Section 17, “Developing System-Level NCL Procedures,” discusses how to develop
system-level NCL procedures in detail.  The complete syntax of the EMS*, LOG*, and
MSG* verbs are discussed in the NonStop NET/MASTER NCL Reference Manual.

To show how a MSGPROC NCL procedure works, this subsection discusses two NCL
procedures.  The first NCL procedure, the MESSGEN NCL procedure, generates
messages that are sent to your OCS window.  The second NCL procedure, the
MSGPROC NCL procedure, filters and processes these messages before they arrive at
the OCS window.

To appreciate how the two procedures interact—how the MSGPROC NCL procedure
filters and processes messages from the MESSGEN NCL procedure—you must create
the procedures separately and then execute them in the correct sequence.  The
following five steps summarize what you need to do to appreciate the affect of the
MSGPROC NCL procedure when it is installed:

1. Create, test compile, correct, and execute the MESSGEN NCL procedure.
Carefully examine the output displayed at your OCS window because the
MSGPROC NCL procedure filters and modifies some of the output when
MSGPROC is installed.

2. Terminate execution of the MESSGEN NCL procedure by exiting from OCS.

3. Create, test compile, correct, and execute the MSGPROC NCL procedure.  Ensure
that you use the PROFILE MSGPROC command to execute MSGPROC because
NonStop NET/MASTER MS displays an error if you attempt to execute
MSGPROC using either the START or EXEC command.

Note You must use the PROFILE MSGPROC command because MSGPROC is part of your OCS profile.  You
require an authority level of 2 or more, by default, to execute the PROFILE MSGPROC command.

4. Execute the MESSGEN NCL procedure again in the same window from which you
executed the MSGPROC NCL procedure.  Carefully contrast the output now
displayed at your OCS window with the output displayed earlier before the
MSGPROC NCL procedure was installed.

5. Terminate execution of both the MSGPROC and MESSGEN NCL procedures by
exiting from OCS.

The MESSGEN NCL
Procedure

The first NCL procedure we will create generates messages that are sent to the OCS
window from which the procedure is executed.  The name of the NCL procedure is
MESSGEN:

 messgen: PROCEDURE
    /* A message generator */
    DO FOREVER
       WRITE DATA=Message
       WRITE DATA=Hello message
       WRITE DATA=Hello message &SYS.USER.ID
       DELAY 10
    END
 END messgen

Follow the steps described earlier in this section to create, test compile, correct, and
execute MESSGEN.  The following screen shows the results of executing the
MESSGEN NCL procedure in the OCS message display area:

  (13:02) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START MESSGEN
   Message
   Hello message
   Hello message NNMUSER
   Message
   Hello message
   Hello message NNMUSER
   Message
   Hello message
   Hello message NNMUSER
   Message
   Hello message
   Hello message NNMUSER
   Message
   Hello message
   Hello message NNMUSER
   Message
   Hello message
   Hello message NNMUSER
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Examine this screen carefully, because it is different when you use a MSGPROC NCL
procedure to filter messages before they arrive at the OCS window.

Note After you have examined the output, terminate execution of MESSGEN by using the PROFILE
MSGPROC=FLUSH command or by typing X in the OCS command line to exit from OCS.

Features of MESSGEN Here are some features of the MESSGEN NCL procedure.

The DO FOREVER Core Statement

DO marks the beginning of a group of statements that is executed once or repetitively.
If DO marks the beginning of a repetitive DO loop, the loop is executed the number of
times specified by a condition.  In MESSGEN, the FOREVER keyword sets the
condition.

(Although the FOREVER keyword implies that the DO loop is executed forever, this is
not so in practice;  you can instruct NCL to control execution of a runaway DO loop.
See Section 5, “Controlling Execution Flow,” for more information.)

The &SYS.USER.ID System Variable

The system variable &SYS.USER.ID holds the user ID of the user executing the NCL
procedure.

Note Your user ID may be different from the user ID, NNMUSER, shown in this example.

The DELAY Verb

DELAY temporarily suspends execution of an NCL process.  In MESSGEN, the NCL
process is suspended for 10 seconds.

The END Core Statements

The first END core statement ends the DO loop.  The last END core statement ends the
NCL procedure.

The MSGPROC NCL
Procedure

The MSGPROC NCL procedure, shown in the following listing, filters and processes
messages before they arrive at the OCS window:

 msgproc: PROCEDURE
    /* Simple MSGPROC procedure */
    DO FOREVER
    /* Read message text into variables */
       MSGREAD VARS=&word*
    /* Process message */
       IF &word1 = "Hello" THEN
          IF &word&(SYS.VARCNT) \= &SYS.USER.ID THEN
    /* Delete if begins with Hello and not ours */
             MSGDEL
          ELSE
    /* Modify if begins with Hello and ours */
             MSGCONT COLOR=BLUE HLITE=BLINK
       ELSE
    /* No change if does not begin with Hello */
          MSGCONT
    END
 END msgproc

Follow the steps described earlier in this section to create, test compile, and correct
MSGPROC.  However, do not use the START or EXEC command to execute the
MSGPROC NCL procedure.  To install the procedure as part of your OCS profile, you
must enter the following command from the OCS command input line:

PROFILE MSGPROC=MSGPROC

After you install the procedure, you can check the results.  The following screen shows
the results of installing the MSGPROC NCL procedure as part of your OCS profile:

  (13:10) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=MSGPROC
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

This message indicates that the MSGPROC NCL procedure is working in the
background to intercept and process all messages arriving at the OCS window.

You can confirm that you have correctly installed the MSGPROC NCL procedure as
part of your OCS profile by typing the PROFILE command (without any operands)
and examining the display:

PROFILE

The MSGPROC NCL procedure is correctly installed if you see the display highlighted
in the following screen:

  (13:11) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE
   NNM0381 USERID: NNMUSER  TERMINAL-ID: #9246528
   NNM0382 NCL PROCEDURE LIBRARY ID: $DATA9.MYNCLLIB
   NNM0383 USERID HAS MONITOR STATUS.
   NNM0384 AUTOHOLD SETTING IS ON AUTO.
   NNM0385 AUTHORITY LEVEL IS 2, MSGPROC MSGPROC ACTIVE.
   NNM0386 USER SERVICES PROCEDURE: USRSERN
   NNM0388 EMS MESSAGE DELIVERY DETAILS:
   NNM0390 .. RECEIVE ALL EMS MESSAGES.
   NNM0358 Operand        Function
   NNM0359 (CMDKEEP=NO..) NO COMMAND WILL BE KEPT ON ENTRY LINE.
   NNM0361 (CMDSTACK=20.) COMMAND STACK SIZE IS 20.
   NNM0362 (DELCHAR=_...) ROLL DELIMITER LINE CHARACTER IS '_'.
   NNM0364 (HOLD=200....) HELD MESSAGES THRESHOLD IS 200.
   NNM0365 (INITCMD=....) INITIAL CMD IS: AUTOHOLD ON;PROFILE CMDKEEP=NO.
   NNM0378 (EMS=YES.....) EMS MESSAGES WILL BE DISPLAYED. CLASS=NORM.
   NNM0375 (ROLL=1......) SCREEN ROLL DELAY TIME IS 1.
   NNM0366 (MONMSG=YES..) MONITOR CLASS MESSAGES WILL BE DISPLAYED.
   NNM0372 (NRDELCH==...) NRD DELIMITER LINE CHARACTER IS '='.
   NNM0374 (RDMIN=1.....) MINIMUM RDM LINES IS 1.
   NNM0368 (MSGALARM=0..) QUEUED MESSAGE ALARM COUNT IS 0.
 AUTO HOLD  ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Features of MSGPROC Comments throughout the MSGPROC NCL procedure summarize what is happening
at each step.  Here are some additional comments on its features.

The MSGREAD Verb and the VARS Keyword

The MSGREAD verb is used only in a MSGPROC NCL procedure.  It reads the next
message destined for an OCS window.  The VARS keyword reads each word of the
text of the message into the specified variables.  The variables following VARS begin
with &WORD, and NCL automatically creates variables beginning with &WORD—
&WORD1, &WORD2, and so on—to hold all the words in the text.

Note The whole message, the attributes as well as the text, is also read into a mapped data object (MDO)
variable beginning with the reserved stem &$MSG..  MDO variables are discussed in more detail in
Section 10, “Working With Mapping Services.”

The IF Core Statements and the MSGDEL and MSGCONT Verbs

After a message is read, the IF core statements determine how to process the message.
Depending on the text, the IF statements can:

Delete the message using the MSGDEL verb.  The message is not displayed at the
OCS window.

Modify the message and forward it on using the MSGCONT verb.  The modified
message is displayed at the OCS window.

Leave the message unchanged and forward it on using the MSGCONT verb.  The
unchanged message is displayed at the OCS window.

The &WORD&(SYS.VARCNT) Complex, Nested Variable

The &WORD&(SYS.VARCNT) is a complex, nested variable.  A complex variable is a
variable with more than one ampersand (&).  A nested variable is a variable that
begins with an ampersand and an opening parenthesis and terminates with a closing
parenthesis (&, (, and )).  This type of variable allows you to combine a simple
variable, &WORD, with a system variable, &SYS.VARCNT.  The complex, nested
variable used in the MSGPROC NCL procedure is used to determine the value of the
last variable created by the VARS keyword.

MSGPROC and MESSGEN To see how the MSGPROC NCL procedure filters messages, with MSGPROC active
execute the MESSGEN NCL procedure again in the same window from which you
executed MSGPROC.  Carefully contrast the output now displayed at your OCS
window with the output displayed earlier, before the MSGPROC NCL procedure was
installed.

The following screen shows the results of executing MESSGEN with MSGPROC
active:

  (13:33) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=MSGPROC
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START MESSGEN
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   Message
   Hello message NNMUSER
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Note The screen uses bold italics to indicate a blue, blinking message.

The preceding screen shows that the MSGPROC NCL procedure does the following:

Deletes the “Hello message” message—it begins with “Hello” but does not end
with your user ID.

Modifies the attributes of the “Hello message NNMUSER” message—it begins
with “Hello” and ends with your user ID—and forwards it on to the OCS window.

Note Your user ID may be different from the user ID, NNMUSER, shown in this example.

Leaves the “Message” message unchanged—it does not begin with “Hello”—and
forwards it on to the OCS window.

Note After you have examined the results of the MSGPROC NCL procedure, you can terminate execution of
both the MSGPROC and MESSGEN NCL procedures by exiting from OCS.

What to Do Next This concludes the NCL tutorial.  To continue to use this manual in the most effective
way possible, you should now read Sections 4 through 8.  These sections discuss more
of the details you need to understand before you begin to develop more complex NCL
procedures.  The sections discuss initial procedure development, the core statements
used to organize the structure of NCL procedures and user-written functions, and the
many techniques you can use to execute NCL procedures.

If you want to know more about all topics covered in individual sections of this
manual, you should read “About This Manual” again before you continue.

